NAG Library Routine Document

F07BTF (ZGBEQU)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07BTF (ZGBEQU) computes diagonal scaling matrices D_R and D_C intended to equilibrate a complex m by n band matrix A of band width $(k_l + k_u + 1)$, and reduce its condition number.

2 Specification

```
SUBROUTINE F07BTF (M, N, KL, KU, AB, LDAB, R, C, ROWCND, COLCND, AMAX,     &
                   INFO)

INTEGER                 M, N, KL, KU, LDAB, INFO
REAL (KIND=nag_wp)      R(M), C(N), ROWCND, COLCND, AMAX
COMPLEX (KIND=nag_wp) AB(LDAB,*)
```

The routine may be called by its LAPACK name ***zgbequ***.

3 Description

F07BTF (ZGBEQU) computes the diagonal scaling matrices. The diagonal scaling matrices are chosen to try to make the elements of largest absolute value in each row and column of the matrix B given by

$$B = D_R A D_C$$

have absolute value 1. The diagonal elements of D_R and D_C are restricted to lie in the safe range $(\delta, 1/\delta)$, where δ is the value returned by routine X02AMF. Use of these scaling factors is not guaranteed to reduce the condition number of A but works well in practice.

4 References

None.

5 Arguments

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: M ≥ 0.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: N ≥ 0.

3: KL – INTEGER *Input*

On entry: k_l, the number of subdiagonals of the matrix A.

Constraint: KL ≥ 0.

4: KU – INTEGER *Input*

On entry: k_u, the number of superdiagonals of the matrix A.

Constraint: KU ≥ 0.

5: AB(LDAB, $*$) – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array AB must be at least max(1, N).

On entry: the m by n band matrix A whose scaling factors are to be computed.

The matrix is stored in rows 1 to $k_l + k_u + 1$, more precisely, the element A_{ij} must be stored in

$$\text{AB}(k_u + 1 + i - j, j) \quad \text{for } \max(1, j - k_u) \leq i \leq \min(m, j + k_l).$$

See Section 9 in F07BNF (ZGBSV) for further details.

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07BTF (ZGBEQU) is called.

Constraint: LDAB $\geq$ KL + KU + 1.

7: R(M) – REAL (KIND=nag_wp) array *Output*

On exit: if INFO $= 0$ or INFO $>$ M, R contains the row scale factors, the diagonal elements of D_R. The elements of R will be positive.

8: C(N) – REAL (KIND=nag_wp) array *Output*

On exit: if INFO $= 0$, C contains the column scale factors, the diagonal elements of D_C. The elements of C will be positive.

9: ROWCND – REAL (KIND=nag_wp) *Output*

On exit: if INFO $= 0$ or INFO $>$ M, ROWCND contains the ratio of the smallest value of R(i) to the largest value of R(i). If ROWCND ≥ 0.1 and AMAX is neither too large nor too small, it is not worth scaling by D_R.

10: COLCND – REAL (KIND=nag_wp) *Output*

On exit: if INFO $= 0$, COLCND contains the ratio of the smallest value of C(i) to the largest value of C(i).

If COLCND ≥ 0.1, it is not worth scaling by D_C.

11: AMAX – REAL (KIND=nag_wp) *Output*

On exit: $\max |a_{ij}|$. If AMAX is very close to overflow or underflow, the matrix A should be scaled.

12: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0 and INFO $\leq$ M

Row $\langle value \rangle$ of A is exactly zero.

INFO $>$ M

Column $\langle value \rangle$ of A is exactly zero.

7 Accuracy

The computed scale factors will be close to the exact scale factors.

8 Parallelism and Performance

F07BTF (ZGBEQU) is not threaded in any implementation.

9 Further Comments

The real analogue of this routine is F07BFF (DGBEQU).

10 Example

This example equilibrates the complex band matrix A given by

$$A = \begin{pmatrix} -1.65 + 2.26i & (-2.05 - 0.85i) \times 10^{-10} & 0.97 - 2.84i & 0 \\ 0.00 + 6.30i & (-1.48 - 1.75i) \times 10^{-10} & -3.99 + 4.01i & 0.59 - 0.48i \\ 0 & -0.77 + 2.83i & (-1.06 + 1.94i) \times 10^{10} & (3.33 - 1.04i) \times 10^{10} \\ 0 & 0 & 0.48 - 1.09i & -0.46 - 1.72i \end{pmatrix}.$$

Details of the scaling factors, and the scaled matrix are output.

10.1 Program Text

```
    Program f07btfe

!     F07BTF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: f06kcf, nag_wp, x02ajf, x02amf, x02bhf, x04def,   &
                             zdscal, zgbequ
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Real (Kind=nag_wp), Parameter    :: one = 1.0_nag_wp
      Real (Kind=nag_wp), Parameter    :: thresh = 0.1_nag_wp
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: amax, big, colcnd, rowcnd, small
      Integer                          :: i, i0, i1, ifail, ilen, info, j, k,  &
                                          kl, ku, ldab, n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,:)
      Real (Kind=nag_wp), Allocatable  :: c(:), r(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min, real
!     .. Executable Statements ..
      Write (nout,*) 'F07BTF Example Program Results'
      Write (nout,*)
      Flush (nout)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kl, ku
      ldab = kl + ku + 1
```

```
      Allocate (ab(ldab,n),c(n),r(n))

!     Read the band matrix A from data file

      k = ku + 1
      Read (nin,*)((ab(k+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,n)

!     Print the matrix A

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04def(n,n,kl,ku,ab,ldab,'Matrix A',ifail)

      Write (nout,*)

!     Compute row and column scaling factors

!     The NAG name equivalent of zgbequ is f07btf
      Call zgbequ(n,n,kl,ku,ab,ldab,r,c,rowcnd,colcnd,amax,info)

      If (info>0) Then
        If (info<=n) Then
          Write (nout,99999) 'Row ', info, ' of A is exactly zero'
        Else
          Write (nout,99999) 'Column ', info - n, ' of A is exactly zero'
        End If
      Else

!       Print ROWCND, COLCND, AMAX and the scale factors

        Write (nout,99998) 'ROWCND =', rowcnd, ', COLCND =', colcnd,          &
          ', AMAX =', amax
        Write (nout,*)
        Write (nout,*) 'Row scale factors'
        Write (nout,99997) r(1:n)
        Write (nout,*)
        Write (nout,*) 'Column scale factors'
        Write (nout,99997) c(1:n)
        Write (nout,*)
        Flush (nout)

!       Compute values close to underflow and overflow

        small = x02amf()/(x02ajf()*real(x02bhf(),kind=nag_wp))
        big = one/small
        If ((rowcnd>=thresh) .And. (amax>=small) .And. (amax<=big)) Then
          If (colcnd<thresh) Then

!           Just column scale A
!           The NAG name equivalent of zdscal is f06jdf
            Do j = 1, n
              i1 = 1 + max(1,j-ku) - (j-ku)
              ilen = min(n,j+kl) - max(1,j-ku) + 1
              Call zdscal(ilen,c(j),ab(i1,j),1)
            End Do

          End If
        Else If (colcnd>=thresh) Then

!         Just row scale A
          Do j = 1, n
            i0 = max(1,j-ku)
            i1 = 1 + i0 - (j-ku)
            ilen = min(n,j+kl) - i0 + 1
            Call f06kcf(ilen,r(i0),1,ab(i1,j),1)
          End Do

        Else

!         Row and column scale A
```

```
          Do j = 1, n
            i0 = max(1,j-ku)
            i1 = 1 + i0 - (j-ku)
            ilen = min(n,j+kl) - i0 + 1
            Call zdscal(ilen,c(j),ab(i1,j),1)
            Call f06kcf(ilen,r(i0),1,ab(i1,j),1)
          End Do

        End If

!       Print the scaled matrix
        ifail = 0
        Call x04def(n,n,kl,ku,ab,ldab,'Scaled matrix',ifail)

      End If

99999 Format (1X,A,I4,A)
99998 Format (1X,3(A,1P,E8.1))
99997 Format ((1X,1P,7E11.2))
    End Program f07btfe
```

10.2 Program Data

```
F07BTF Example Program Data
  4  1  2                                                    :Values of N, KL and KU
 (-1.65, 2.26) (-2.05D-10,-8.50D-11) ( 9.70D-01,-2.84D+00)
 ( 0.00, 6.30) (-1.48D-10,-1.75D-10) (-2.99D+00, 3.01D+00) ( 0.59D+00,-0.48D+00)
               (-7.70D-01, 2.83D+00) (-1.06D+10, 1.94D+10) ( 3.33D+10,-1.04D+10)
                                     ( 4.48D+00,-1.09D+00) (-0.46D+00,-1.72D+00)
                                                           :End of matrix A
```

10.3 Program Results

```
 F07BTF Example Program Results

 Matrix A
                1            2            3            4
 1   -1.6500E+00  -2.0500E-10   9.7000E-01
      2.2600E+00  -8.5000E-11  -2.8400E+00

 2    0.0000E+00  -1.4800E-10  -2.9900E+00   5.9000E-01
      6.3000E+00  -1.7500E-10   3.0100E+00  -4.8000E-01

 3               -7.7000E-01  -1.0600E+10   3.3300E+10
                  2.8300E+00   1.9400E+10  -1.0400E+10

 4                             4.4800E+00  -4.6000E-01
                              -1.0900E+00  -1.7200E+00

 ROWCND = 8.9E-11, COLCND = 8.2E-11, AMAX = 4.4E+10

 Row scale factors
    2.56E-01   1.59E-01   2.29E-11   1.80E-01

 Column scale factors
    1.00E+00   1.21E+10   1.00E+00   1.00E+00

 Scaled matrix
          1       2       3       4
 1  -0.4220 -0.6364  0.2481
     0.5780 -0.2639 -0.7263

 2   0.0000 -0.2852 -0.4746  0.0937
     1.0000 -0.3372  0.4778 -0.0762
```

```
 3          -0.2139 -0.2426  0.7620
             0.7861  0.4439 -0.2380

 4                   0.8043 -0.0826
                    -0.1957 -0.3088
```